

## David Ghan (suite)

**Animaux** : Nous avons une chatte tachetée de 4 ans répondant au nom de Toby.

**Sports/Passe-temps** : J'aime jouer de la guitare et de la viole. J'aime aussi jouer au hockey tôt le samedi matin avec un groupe d'amis (quand j'arrive à me lever, il va s'en dire).

**Une fin de semaine idéale** : Après une bonne nuit de sommeil, se réveiller au bord d'un lac dans le parc Algonquin par une belle journée ensoleillée, se faire un petit déjeuner avec bacon et crêpes, puis passer la journée à faire du canot, à nager et à pêcher avec ma famille. Sensiblement la même chose le dimanche, avec œufs et rôties au menu.

**Mets favoris** : Salade grecque, viande grillée, tarte aux fruits. Généralement, tout ce qui se trouve dans mon assiette.

**Si je pouvais être quelqu'un d'autre (qu'un programmeur SAS), je voudrais être ....** une star du rock.

**Lorsque je ne fais pas de programmation SAS, j'aime bien .....** faire semblant d'être une star du rock. En fait, j'aime bien faire semblant d'être une star du rock LORSQUE je fais de la programmation SAS.

**Une chose que tout programmeur SAS devrait savoir....**

### Utiliser Proc Template pour contrôler la disposition de la sortie de la procédure

On commence en créant une sortie HTML pour Proc Means :

```
ods html file='means sol.html' style=sasweb ;
proc means data=ia.employees n min mean max p10 p25 p50 p75 p90 ;
  var salary;
  class JobCode;
  where jobcode contains 'FLT';
run;
ods html close;
```

La sortie est très étendue à cause de toutes les statistiques :

Analysis Variable : Salary Employee Salary										
Job Code	N Obs	N	Minimum	Mean	Maximum	10th Pctl	25th Pctl	50th Pctl	75th Pctl	90th Pctl
FLTAT1	170	170	16000.00	29594.12	45000.00	19000.00	23000.00	29000.00	37000.00	41000.00
FLTAT2	227	227	16000.00	30691.63	45000.00	19000.00	22000.00	31000.00	38000.00	43000.00
FLTAT3	354	354	16000.00	30861.58	45000.00	20000.00	23000.00	31000.00	39000.00	43000.00

Nous allons diminuer la largeur du tableau et l'allonger en empilant les statistiques. Cela permet de faire cadrer toute l'information dans les limites permises d'un document imprimé.

Voici le type de sortie que nous allons créer :

Salary Statistics		
Job Code	Employees	Statistic
FLTAT1	170	min= 16000 mean= 29594 max= 45000 10th pctl= 19000 25th pctl= 23000 median= 29000 75th pctl= 37000 90th pctl= 41000
FLTAT2	227	min= 16000 mean= 30691 max= 45000 10th pctl= 19000 25th pctl= 22000 median= 31000 75th pctl= 38000 90th pctl= 43000
FLTAT3	354	min= 16000 mean= 30861 max= 45000 10th pctl= 20000 25th pctl= 23000 median= 31000 75th pctl= 39000 90th pctl= 43000

Remarquez que les 8 statistiques ont été empilées dans une seule colonne au lieu d'être présentées dans des colonnes séparées. De plus, un format d'étiquette est utilisé pour afficher le nom de chaque statistique à gauche de la valeur statistique.

La sortie ci-dessus est encore produite au moyen de Proc Means, mais en modifiant le **modèle de table** qu'utilise SAS pour déterminer la disposition. La plupart des procédures SAS utilisent des modèles prédéfinis de table qui déterminent comment chaque objet de sortie est affiché. Ces définitions peuvent être modifiées pour s'adapter à la disposition souhaitée par l'utilisateur.

Pour notre objectif, la première tâche consiste à déterminer le nom du modèle de table pour Proc Means à l'aide de ODS Trace :

```
ods trace on / label;  
proc means data=ia.employees n min mean max p10 p25 p50 p75 p90 ;  
  var salary;  
  class JobCode;  
  where jobcode contains 'FLT';  
run;  
ods trace off;
```

## Le journal SAS :

Output Added:

```
-----  
Name:      Summary  
Label:     Summary statistics  
Template:  base.summary  
Path:      Means.Summary  
Label Path: 'The Means Procedure'. 'Summary statistics'
```

---

Nous pouvons modifier ce modèle avec la procédure template :

```
proc template;  
  edit Base.Summary;  
    column class n (min mean max p10  
                  p25 p50 p75 p90)  
    ;  
  header h;  
  define h;  
    text "Salary Statistics";  
  end;  
  edit min;  
    header = "Statistic";  
    format = minfmt.;  
    generic;  
    just=R;  
  end;  
  edit max;  
    format = maxfmt.;  
    generic;  
    just=R;  
  end;  
  edit n;  
    header= "Employees";  
    format = comma8.;  
    vjust=top;  
    generic;  
  end;  
  edit mean;  
    format = meanfmt.;  
    generic;  
    just=R;  
  end;  
  edit p10;  
    just=R;  
    format=p10fmt17.;  
  end;  
  edit p25;  
    just=R;  
    format=p25fmt17.;  
  end;  
  . . .  
  . . .  
end;  
run;
```

L'instruction *column* empile toutes les statistiques de min à p90 à la verticale dans une seule colonne en les encadrant par des parenthèses

Salary Statistics		
Job Code	Employees	Statistic
FLTAT1	170	min= 16000 mean= 29594 max= 45000 10th pctl= 19000 25th pctl= 23000 median= 29000 75th pctl= 37000 90th pctl= 41000
FLTAT2	227	min= 16000 mean= 30691 max= 45000 10th pctl= 19000 25th pctl= 22000 median= 31000

Remarquez également l'utilisation des formats d'étiquette pour identifier chaque statistique dans la table

Le format d'étiquette a été créé au moyen de Proc Format :

```
proc format;
  picture minfmt low-high=00009 (prefix='min= ');
  picture meanfmt low-high=00009 (prefix='mean= ');
  picture maxfmt low-high=00009 (prefix='max= ');
  picture p10fmt low-high=00009 (prefix='10th pctl= ');
  picture p25fmt low-high=00009 (prefix='25th pctl= ');
  picture p50fmt low-high=00009 (prefix='median= ');
  picture p75fmt low-high=00009 (prefix='75th pctl= ');
  picture p90fmt low-high=00009 (prefix='90th pctl= ');
run;
```

Les formats d'étiquette donnent des directives sur la façon d'afficher les nombres précédés de caractères, imbriqués à la valeur numérique ou à sa suite.

Le programme complet du début à la fin se lit comme suit :

```
ods path work.temptemp(update)
      sasuser.templat(update)
      sashelp.tmplmst(read);

ods listing close;
proc format;
  picture minfmt low-high=00009 (prefix='min= ');
  picture meanfmt low-high=00009 (prefix='mean= ');
  picture maxfmt low-high=00009 (prefix='max= ');
  picture p10fmt low-high=00009 (prefix='10th pctl= ');
  picture p25fmt low-high=00009 (prefix='25th pctl= ');
  picture p50fmt low-high=00009 (prefix='median= ');
  picture p75fmt low-high=00009 (prefix='75th pctl= ');
  picture p90fmt low-high=00009 (prefix='90th pctl= ');
run;
proc template;
  edit Base.Summary;
    column class n (min mean max p10 p25 p50 p75 p90) ;
    header h;
    define h;
      text "Salary Statistics";
    end;
  edit min;
    header = "Statistic";
    format = minfmt.;
    generic;
    just=R;
  end;
  edit max;
    format = maxfmt.;
    generic;
    just=R;
  end;
  edit n;
    header= "Employees";
```

```

        format = comma8.;
        vjust=top;
        generic;
    end;
    edit mean;
        format = meanfmt.;
        generic;
        just=R;
    end;
    edit p10;
        just=R;
        format=p10fmt17.;
    end;
    edit p25;
        just=R;
        format=p25fmt17.;
    end;
    edit p50;
        just=R;
        format=p50fmt17.;
    end;
    edit p75;
        just=R;
        format=p75fmt17.;
    end;
    edit p90;
        just=R;
        format=p90fmt17.;
    end;
end;
run;

ods html file='means_sol.html' style=sasweb ;
title;
proc means data=ia.employees n min mean max p10 p25 p50 p75 p90 ;
    var salary;
    class JobCode;
    where jobcode contains 'FLT';
run;
ods html close;
ods listing;

```